



МИНОБРНАУКИ РОССИИ

Федеральное государственное бюджетное образовательное учреждение  
высшего образования

«МИРЭА – Российский технологический университет»

**РТУ МИРЭА**

## Методические указания к выполнению практических работ

### Моделирование информационно-аналитических систем

#### Практическая работа 8

	<i>(наименование дисциплины (модуля) в соответствии с учебным планом)</i>		
Уровень	специалитет		
	<i>(бакалавриат, магистратура, специалитет)</i>		
Форма обучения	очная		
	<i>(очная, очно-заочная, заочная)</i>		
Направление(-я) подготовки	10.05.04	Информационно-аналитические системы безопасности, специализации: специализация №1 "Автоматизация информационно-аналитической деятельности"; специализация №3 "Технологии информационно-аналитического мониторинга".	
	<i>(код(-ы) и наименование(-я))</i>		
Институт	Кибербезопасности и цифровых технологий		
	<i>(полное и краткое наименование)</i>		
Кафедра	Информационно-аналитические системы кибербезопасности (КБ-2)		
	<i>(полное и краткое наименование кафедры, реализующей дисциплину (модуль))</i>		
Лектор	к.т.н., доцент Лебедев Владимир Владимирович		
	<i>(сокращенно – ученая степень, ученое звание; полностью – ФИО)</i>		
Используются в данной редакции с учебного года	2024/25		
	<i>(учебный год цифрами)</i>		
Проверено и согласовано «___» _____ 20__ г.			
	<i>(подпись директора Института/Филиала с расшифровкой)</i>		

Москва 20\_\_ г.

## Практическое занятие №8


### **Численное имитационное моделирование стохастической системы обработки заявок на предоставление ресурса в автоматизированной информационной системе с отказами. Модели DOS-атак.**

#### ***СМО с ограниченной ёмкостью накопителя с одним каналом обслуживания***

Если в системе с одним каналом имеет место ограниченное время ожидания или ограниченное число мест в очереди, то также может возникать поток отказов для заявок, которые удаляются из системы при превышении ими времени пребывания в очереди или отсутствии мест в ней.

Рассмотрим модель СМО с одним каналом, ограниченным числом мест в очереди и потоком отказов.

Граф модели Марковского процесса в СМО с одним каналом, ограниченным числом мест в очереди и потоком отказов представлен ниже на рис. 1.



*Рисунок 1. Граф Марковской модели СМО с  
ограниченным числом мест в очереди и одним каналом*

#### **Задание.**

1. Разработать модель с одним каналом и ограниченным числом мест в накопителе. Построить зависимость количества отказов от времени при заданных параметрах модели: число мест в очереди, интенсивности поступления заявок в систему  $\lambda$  и интенсивности их обслуживания  $\mu$ .

2. Привести текст программы расчёта.

Варианты для выполнения расчётов заданы с использованием генератора

заданий.

3. Построить графики изменения количества отказов в обслуживании от времени процесса по временной шкале событий, связанных с занятием и освобождением каналов обслуживания.

Сделать выводы.

#### **Пояснения к выполнению задания.**

1. Сформировать временные шкалы событий в системе:

1-ая шкала – это случайные моменты времени занятия очереди;

2-ая шкала – это случайные моменты времени освобождения очереди (передача транзактов на обслуживание), для расчёта этих моментов в модели использовать формулу:  $t_k^{нач\ обсл} = t_k^{пост} + \Delta t_k^{ож} ..$

Для моделирования временных шкал создать модель журнала событий (объём выборки 500 событий) в СМО *с одним каналом и неограниченным числом мест в очереди*. В журнале фиксируется момент входа транзакта в систему – 1-я шкала, момент передачи транзакта на обслуживание – 2-я шкала и момент его выхода из системы по окончании обслуживания – 3-я шкала.

2. Сформировать таблицу на основе событий журнала объёмом 500. Для этого расширить шкалы времени событий добавлением столбца с количеством “1” входящих в систему транзактов для 1-ой шкалы (занятие очереди) и ”-1” для моментов времени передачи транзактов на обслуживание (освобождение очереди). Объединить расширенные шкалы в таблицу.

3. Создать новую таблицу. Для этого объединить расширенные шкалы слиянием по принципу «смещения» соседних событий из 1-й и 2-й шкал. Объём событий в новой таблице удвоится до 1000.

4. Произвести сортировку таблицы по возрастанию значений столбца “моменты времени”, используя “пузырьковый” алгоритм сортировки.

5. Добавить справа к таблице столбец, в котором выполнить накопительное суммирование значений столбца “Выделяемые/освобождаемые места в очереди”, начиная с верхней ячейки.

В этом столбце будет отражаться текущее количество занятых мест в очереди.

Построить диаграмму динамики процесса количества занимаемых мест в очереди в разные моменты времени по объединённой шкале.

На графике диаграммы изобразить красную линию ограничения мест в накопителе: число мест равно 40.

Пики случайных значений числа мест заявок – требований на занятие мест в очереди в некоторые моменты времени рассматриваемого дискретного процесса выходят над этой линией. В эти моменты в системе с ограниченным накопителем возникают отказы.

6. Задав конечное число мест в очереди (*ограниченная ёмкость накопителя* – вариант задания), смоделировать поток отказов.

Модель построена с применением алгоритма расщепления событий в сформированной шкале событий в первой модели без ограничений. При превышении суммы текущих заявок над количеством занятых мест в накопителе (очереди на обслуживание) происходит фиксация событий потока отказов, при котором заявка, которая пришла систему, когда все места в очереди заняты, снимается с обслуживания.

Пример текста программы:

```
Private Sub Cr_un_shkala_sort_()  
'Модель СМО с одним каналом и накопителем  
'Динамика заявок на занятие мест в накопителе  
'Динамика потока отказов при ограниченной ёмкости накопителя  
  
Dim K As Integer, Nque As Integer, Lyam, Mu  
Dim A1(), A2(), A3(), A4(), Adet(), Assin()  
Dim A5(), A6(), A7(), A8(), A2_out()  
  
'A1 - шкала моментов времени поступления заявок в СМО  
'Adet - задержка заявок в очереди  
'A2 - шкала моментов времени передачи заявок из очереди на обслуживание  
'A3 - сводная шкала событий в накопителе  
'A4 - сортировка сводной шкалы событий  
'A5 - динамика заявок на занятие мест в накопителе  
'A6 - шкала потока отказов  
'A7 - модифицированная шкала событий  
'K - число заявок в цепи событий
```

```

'Nque - число мест в накопителе в СМО
'Lyam - интенсивность потока заявок
'Mu - интенсивность потока обслуживания

'Ввод исходных данных
K = InputBox("Ввести длину цепи событий в СМО")
Lyam = InputBox("Ввести интенсивность потока событий")
Mu = InputBox("Ввести интенсивность потока обслуживания")
Nque = InputBox("Ввести число мест в накопителе в СМО")

ReDim A1(1 To K, 1 To 2), A2(1 To K, 1 To 2), Adet(1 To K, 1 To 2), A2_out(1 To K, 1 To 2), Assin(1 To K, 1 To 2)
ReDim A3(1 To 2 * K, 1 To 3), A4(1 To 2 * K, 1 To 3), A5(1 To 2 * K, 1 To 3)
ReDim A6(1 To 2 * K, 1 To 3), A7(1 To 2 * K, 1 To 3), A8(1 To 2 * K, 1 To 3)

'Ввод шкалы моментов времени поступления заявок на вход СМО
For J = 1 To K
    Randomize
    r = Rnd()
    DTin = (1 / Lyam) * Log(1 / (1 - r))
    If J = 1 Then
        A1(J, 1) = DTin
    Else
        A1(J, 1) = A1(J - 1, 1) + DTin
    End If
    A1(J, 2) = 1
Next J

'Формирование шкал моментов времени передачи запросов из очереди на обслуживание
и моментов времени окончания обслуживания
'Расчёт интервалов времени на обслуживание, ожидание в очереди и пребывание в АИС
For J = 1 To K
    Randomize
    r = Rnd()
    DTserv = (1 / Mu) * Log(1 / (1 - r))
    If J = 1 Then
        Adet(J, 1) = 0
        A2(J, 1) = A1(J, 1) + Adet(J, 1)
        A2_out(J, 1) = A2(J, 1) + DTserv
    Else
        If A2_out(J - 1, 1) > A1(J, 1) Then
            Adet(J, 1) = A2_out(J - 1, 1) - A1(J, 1)
        Else
            Adet(J, 1) = 0
        End If
        A2(J, 1) = A1(J, 1) + Adet(J, 1)
        A2_out(J, 1) = A2(J, 1) + DTserv
    End If
    Adet(J, 2) = J
    If Adet(J, 1) = 0 Then
        A1(J, 2) = 0
    Else
        A1(J, 2) = 1
    End If
    If Adet(J, 1) = 0 Then
        A2(J, 2) = 0
    Else
        A2(J, 2) = -1
    End If
    Assin(J, 1) = A2_out(J, 1) - A1(J, 1)
Next J

```

```

'Формирование сводной шкалы событий
  For M = 1 To K
    A3(2 * M - 1, 1) = A1(M, 1)
    A3(2 * M - 1, 2) = A1(M, 2)
    A3(2 * M - 1, 3) = M
    A3(2 * M, 1) = A2(M, 1)
    A3(2 * M, 2) = A2(M, 2)
    A3(2 * M, 3) = M
  Next M
'Сортировка сводной шкалы событий
  For J = 1 To 2 * K
    A4(J, 1) = A3(J, 1)
    A4(J, 2) = A3(J, 2)
    A4(J, 3) = A3(J, 3)
  Next J
  KK = 1
  Do While KK <= 2 * K
    For J = KK + 1 To 2 * K
      If A4(J, 1) < A4(KK, 1) Then
        AA = A4(KK, 1)
        BB = A4(KK, 2)
        CC = A4(KK, 3)
        A4(KK, 1) = A4(J, 1)
        A4(KK, 2) = A4(J, 2)
        A4(KK, 3) = A4(J, 3)
        A4(J, 1) = AA
        A4(J, 2) = BB
        A4(J, 3) = CC
      End If
    Next J
    KK = KK + 1
  Loop
'Моделирование динамики занятия и освобождения мест в накопителе и формирование
потока отказов
'Динамика требований на занятие мест в накопителе
  Nk = 0
  For M = 1 To 2 * K
    Nk = Nk + A4(M, 2)
    A5(M, 1) = A4(M, 1)
    A5(M, 2) = Nk
    A5(M, 3) = A4(M, 3)
  Next M
'Модифицированная шкала событий при ограниченной ёмкости накопителя СМО и шкала
потока отказов
  For M = 1 To 2 * K
    A7(M, 1) = A4(M, 1)
    A7(M, 2) = A4(M, 2)
    A7(M, 3) = A4(M, 3)
  Next M
  Z = 0
  A6(1, 1) = A7(1, 1)
  A6(1, 2) = 0
  A6(1, 3) = 0
  For M = 2 To 2 * K
    Nkr = 0
    For J = 1 To M
      Nkr = Nkr + A7(J, 2)
    Next J
    If Nkr > Nque Then
      A6(M, 2) = 1
      A6(M, 1) = A7(M, 1)
      A6(M, 3) = A7(M, 3)
      A7(M, 2) = 0
    End If
  Next M

```

```

        Z = A7(M, 3)
        n = M
        E = 0
        Do While E = 0
            n = n + 1
            If A7(n, 3) = Z Then
                T = n
                E = 1
            End If
        Loop
        A7(T, 2) = 0
    Else
        A6(M, 2) = 0
        A6(M, 1) = A7(M, 1)
        A6(M, 3) = 0
        If A4(M, 3) = Z Then
            A7(M, 2) = 0
        End If
        T = M
    End If
Next M
'Динамика занятия мест в накопителе в СМО с ограниченной ёмкостью накопителя
Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A7(M, 2)
    A8(M, 1) = A7(M, 1)
    A8(M, 2) = Nk
    A8(M, 3) = A7(M, 3)
Next M

'Вывод результатов расчёта
ncol = 2
nstr = 20

Ndenide = 0
For J = 1 To 2 * K
    Ndenide = Ndenide + A6(J, 2)
    Cells(nstr + K + 5 + J, ncol + 15) = Ndenide
Next J

Cells(3, 1) = "Длина цепи событий"
Cells(3, 2) = K
Cells(5, 1) = "Lyam"
Cells(6, 1) = "Mu"
Cells(5, 2) = Lyam
Cells(6, 2) = Mu
Cells(nstr, ncol) = "№"
Cells(nstr, ncol + 1) = "Tin"
Cells(nstr, ncol + 2) = "Tque"
Cells(nstr, ncol + 3) = "Tout"
Cells(nstr, ncol + 4) = "DTin"
Cells(nstr, ncol + 5) = "DTwait"
Cells(nstr, ncol + 6) = "DTserv"
Cells(nstr, ncol + 7) = "DTlife"

For M = 1 To K
    Cells(nstr + M, ncol) = M
    Cells(nstr + M, ncol + 1) = A1(M, 1)
    Cells(nstr + M, ncol + 2) = A2(M, 1)
    Cells(nstr + M, ncol + 3) = A2_out(M, 1)
    If M = 1 Then
        Cells(nstr + M, ncol + 4) = A1(M, 1)
    Else

```

```

Cells(nstr + M, ncol + 4) = A1(M, 1) - A1(M - 1, 1)
End If
Cells(nstr + M, ncol + 5) = Adet(M, 1)
Cells(nstr + M, ncol + 6) = A2_out(M, 1) - A2(M, 1)
Cells(nstr + M, ncol + 7) = A2_out(M, 1) - A1(M, 1)
Next M

Cells(nstr + K + 5, ncol) = "№"
Cells(nstr + K + 5, ncol + 1) = "Tin"
Cells(nstr + K + 5, ncol + 2) = "n+"
Cells(nstr + K + 5, ncol + 3) = "Tout"
Cells(nstr + K + 5, ncol + 4) = "n-"
Cells(nstr + K + 5, ncol + 5) = "№"
Cells(nstr + K + 5, ncol + 6) = "UnSc"
Cells(nstr + K + 5, ncol + 7) = "n+/-"
Cells(nstr + K + 5, ncol + 8) = "SortUnSc"
Cells(nstr + K + 5, ncol + 9) = "n+/-"
Cells(nstr + K + 5, ncol + 10) = "n_Din"
Cells(nstr + K + 5, ncol + 11) = "Nlim"
Cells(nstr + K + 5, ncol + 12) = "n_Din_lim"
Cells(nstr + K + 5, ncol + 13) = "n_Din_denide"
Cells(nstr + K + 5, ncol + 14) = "№_Din_denide"
Cells(nstr + K + 5, ncol + 15) = "Sum_D_denide"
Cells(nstr + K + 5, ncol + 16) = "Number_denide"
Cells(nstr + K + 5 + 1, ncol + 16) = Ndenide

For M = 1 To K
Cells(nstr + K + 5 + M, ncol) = M
Cells(nstr + K + 5 + M, ncol + 1) = A1(M, 1)
Cells(nstr + K + 5 + M, ncol + 2) = A1(M, 2)
Cells(nstr + K + 5 + M, ncol + 3) = A2(M, 1)
Cells(nstr + K + 5 + M, ncol + 4) = A2(M, 2)
Next M

For M = 1 To 2 * K
Cells(nstr + K + 5 + M, ncol + 5) = M
Cells(nstr + K + 5 + M, ncol + 6) = A3(M, 1)
Cells(nstr + K + 5 + M, ncol + 7) = A3(M, 2)
Cells(nstr + K + 5 + M, ncol + 8) = A4(M, 1)
Cells(nstr + K + 5 + M, ncol + 9) = A4(M, 2)
Cells(nstr + K + 5 + M, ncol + 10) = A5(M, 2)
Cells(nstr + K + 5 + M, ncol + 11) = Nque
Cells(nstr + K + 5 + M, ncol + 12) = A8(M, 2)
Cells(nstr + K + 5 + M, ncol + 13) = A6(M, 2)
Cells(nstr + K + 5 + M, ncol + 14) = A6(M, 3)
Next M

End Sub

```

7. Провести планированный эксперимент для получения откликов зависимости числа отказов от количества мест в очереди при заданных интенсивностях потока обслуживания  $\mu$  и интенсивностях потока транзактов в СМО  $\lambda$ .

Если менять параметры интенсивности потока заявок на входе в СМО и



число мест в очереди, то с использованием разработанной модели можно получить зависимости числа отказов от указанных параметров.

План вычислительного эксперимента представлен в таблице 1.

Таблица №1. План вычислительного эксперимента

$\lambda =$	$\lambda_B$	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	$\mu_B$	Число отказов							
$\lambda =$	$5\lambda_B$	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	$\mu_B$	Число отказов							
$\lambda =$	$20\lambda_B$	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	$\mu_B$	Число отказов							
	<b>Число отказов при относительном уровне отказов в СМО, %</b>	$p\%$	$\frac{pN}{100}$	$\frac{pN}{100}$	$\frac{pN}{100}$	$\frac{pN}{100}$	$\frac{pN}{100}$	$\frac{pN}{100}$	$\frac{pN}{100}$

Где:  $\lambda_B$ ,  $\mu_B$  – интенсивности потока транзактов и потока обслуживания, заданные по варианту;

$N=500$  – длина цепи событий;

$p$  – заданный уровень отказов, %.

По данным расчёта в строках таблицы построить графики изменения числа отказов (откликов модели) от числа мест ожидания в накопителе.

Для расчёта откликов модернизируем программу, создав подпрограмму функцию.

Текст подпрограммы:

```
Function Den_Que_Nums_(X As Integer, Y As Variant, Z As Variant, V As Integer)
As Variant

'Mодель СМО с одним каналом и накопителем
'Динамика заявок на занятие мест в накопителе
'Динамика потока отказов при ограниченной ёмкости накопителя

Dim K As Integer, Nque As Integer, Lyam As Variant, Mu As Variant, Ndenide As
Variant, DTin As Variant
Dim KK As Integer, Nk As Variant, ZZ As Variant, Nkr As Variant, n As Integer, E
As Variant, T As Integer
Dim A1(), A2(), A3(), A4(), A5(), A6(), A7(), A8(), Adet(), A2_out()

'A1 - шкала моментов времени поступления заявок в СМО
'Adet - задержка заявок в очереди
'A2 - шкала моментов времени передачи заявок из очереди на обслуживание
'A3 - сводная шкала событий в накопителе
'A4 - сортировка сводной шкалы событий
```

```

'A5 - динамика заявок на занятие мест в накопителе
'A6 - шкала потока отказов
'A7 - модифицированная шкала событий
'K - число заявок в цепи событий
'Nque - число мест в накопителе в СМО
'Lyam - интенсивность потока заявок
'Mu - интенсивность потока обслуживания

'Ввод исходных данных
  K = X
  Lyam = Y
  Mu = Z
  Nque = V

ReDim A1(1 To K, 1 To 2), A2(1 To K, 1 To 2), A3(1 To 2 * K, 1 To 3), A4(1 To 2
* K, 1 To 3)
ReDim A5(1 To 2 * K, 1 To 3), A6(1 To 2 * K, 1 To 3), A7(1 To 2 * K, 1 To 3),
A8(1 To 2 * K, 1 To 3)
ReDim A2_out(1 To K), Adet(1 To K, 1 To 2)

'Ввод шкалы моментов времени поступления заявок на вход СМО
  For J = 1 To K
    Randomize
    r = Rnd()
    DTin = (1 / Lyam) * Log(1 / (1 - r))
    If J = 1 Then
      A1(J, 1) = DTin
    Else
      A1(J, 1) = A1(J - 1, 1) + DTin
    End If
    A1(J, 2) = 1
  Next J

'Формирование шкал моментов времени передачи запросов из очереди на обслуживание
и моментов времени окончания обслуживания
'Расчёт интервалов времени на ожидание в очереди и обслуживание в АИС
  For J = 1 To K
    Randomize
    r = Rnd()
    DTserv = (1 / Mu) * Log(1 / (1 - r))
    If J = 1 Then
      Adet(J, 1) = 0
      A2(J, 1) = A1(J, 1) + Adet(J, 1)
      A2_out(J) = A2(J, 1) + DTserv
    Else
      If A2_out(J - 1) > A1(J, 1) Then
        Adet(J, 1) = A2_out(J - 1) - A1(J, 1)
      Else
        Adet(J, 1) = 0
      End If
      A2(J, 1) = A1(J, 1) + Adet(J, 1)
      A2_out(J) = A2(J, 1) + DTserv
    End If
    Adet(J, 2) = J
    If Adet(J, 1) = 0 Then
      A1(J, 2) = 0
    Else
      A1(J, 2) = 1
    End If
    If Adet(J, 1) = 0 Then
      A2(J, 2) = 0
    Else
      A2(J, 2) = -1
    End If
  Next J

```

```

End If
Next J

'Формирование сводной шкалы событий
For M = 1 To K
    A3(2 * M - 1, 1) = A1(M, 1)
    A3(2 * M - 1, 2) = A1(M, 2)
    A3(2 * M - 1, 3) = M
    A3(2 * M, 1) = A2(M, 1)
    A3(2 * M, 2) = A2(M, 2)
    A3(2 * M, 3) = M
Next M

'Сортировка сводной шкалы событий
For J = 1 To 2 * K
    A4(J, 1) = A3(J, 1)
    A4(J, 2) = A3(J, 2)
    A4(J, 3) = A3(J, 3)
Next J
KK = 1
Do While KK <= 2 * K
    For J = KK + 1 To 2 * K
        If A4(J, 1) < A4(KK, 1) Then
            AA = A4(KK, 1)
            BB = A4(KK, 2)
            CC = A4(KK, 3)
            A4(KK, 1) = A4(J, 1)
            A4(KK, 2) = A4(J, 2)
            A4(KK, 3) = A4(J, 3)
            A4(J, 1) = AA
            A4(J, 2) = BB
            A4(J, 3) = CC
        End If
    Next J
    KK = KK + 1
Loop

'Моделирование динамики занятия и освобождения мест в накопителе и формирование
потока отказов
'Динамика требований на занятие мест в накопителе
Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A4(M, 2)
    A5(M, 1) = A4(M, 1)
    A5(M, 2) = Nk
    A5(M, 3) = A4(M, 3)
Next M

'Модифицированная шкала событий при ограниченной ёмкости накопителя СМО и шкала
потока отказов
For M = 1 To 2 * K
    A7(M, 1) = A4(M, 1)
    A7(M, 2) = A4(M, 2)
    A7(M, 3) = A4(M, 3)
Next M
ZZ = 0
A6(1, 1) = A7(1, 1)
A6(1, 2) = 0
A6(1, 3) = 0
For M = 2 To 2 * K
    Nkr = 0
    For J = 1 To M
        Nkr = Nkr + A7(J, 2)
    Next J

```

```

Next J
If Nkr > Nque Then
    A6(M, 2) = 1
    A6(M, 1) = A7(M, 1)
    A6(M, 3) = A7(M, 3)
    A7(M, 2) = 0
    ZZ = A7(M, 3)
    n = M
    E = 0
    Do While E = 0
        n = n + 1
        If A7(n, 3) = ZZ Then
            T = n
            E = 1
        End If
    Loop
    A7(T, 2) = 0
Else
    A6(M, 2) = 0
    A6(M, 1) = A7(M, 1)
    A6(M, 3) = 0
    If A4(M, 3) = ZZ Then
        A7(M, 2) = 0
    End If
    T = M
End If
Next M

'Динамика занятия мест в накопителе в СМО с ограниченной ёмкостью накопителя
Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A7(M, 2)
    A8(M, 1) = A7(M, 1)
    A8(M, 2) = Nk
    A8(M, 3) = A7(M, 3)
Next M

'Определение количества отказов за период наблюдения
Ndenide = 0
For J = 1 To 2 * K
    Ndenide = Ndenide + A6(J, 2)
Next J

Den_Queue_Numbs_ = Ndenide

End Function

```

8. Провести планируемый вычислительный эксперимент для построения регрессии, описывающей зависимость числа отказов от числа мест в очереди.

Параметры модели задать по варианту:

$\lambda =$	$5\lambda_B$
$\mu =$	$\mu_B$

План эксперимента принять согласно таблице 3. Использовать 5 серий параллельных экспериментов.

Таблица 3. План эксперимента для построения регрессии.

Число мест \ Серия	5	10	15	20	25	30	35
1							
2							
3							
4							
5							
Среднее по сериям							

Построить график зависимости по данным строки “среднее по сериям”.

Вычисление откликов производим с использованием подпрограммы функции, представленной в пункте 7.

В качестве модели использовать параметрическую функцию вида:

$$y = N \cdot e^{-ax^b}.$$

В формуле:  $y$  – среднее число отказов;  $x$  – число мест в очереди;  $N$  – длина цепи событий;  $a, b$  – параметры модели регрессии.

Модель регрессии можно охарактеризовать как регрессию с закреплённой точкой, описываемую экспоненциальной функцией со степенной функцией показателя степени.

Рассмотрим некоторые результаты расчёта.

Если число мест очереди в СМО с рассмотренными в предыдущем задании параметрами ограничено, допустим числом 15, то на графике диаграммы количества заявок на обслуживание, см. рис. 2, это представлено красной горизонтальной линией. В моменты времени, когда число заявок – требований на обслуживание превышает число 15, возникает поток отказов, см.

рис. 3. В данном случае при интенсивности потока заявок на входе в СМО  $\lambda = 0,15 \text{ с}^{-1}$  и интенсивности потока обработки заявок в каналах  $\mu = 0,09 \text{ с}^{-1}$  поток отказов, красная линия на диаграмме рис. 3, незначительный. Диаграмма накопления числа отказов за период наблюдения за процессом в пределах 200 транзактов на входе в СМО (длина цепи событий) представлена на рис. 4.




Рисунок 2. Диаграмма количества требований на обслуживание транзактов в разные моменты времени объединённой отсортированной шкалы событий  $\lambda = 0,15 \text{ с}^{-1}$  и интенсивности потока обслуживания  $\mu = 0,09 \text{ с}^{-1}$




Рисунок 3. Диаграмма количества занимаемых мест в очереди (не больше 15) и количества отказов при интенсивности потока транзактов  $\lambda = 0,15 \text{ с}^{-1}$  и интенсивности потока обслуживания  $\mu = 0,09 \text{ с}^{-1}$




Рисунок 4. Диаграмма накопления числа отказов при интенсивности потока транзактов  $\lambda = 0,15 \text{ с}^{-1}$  и интенсивности потока обслуживания  $\mu = 0,09 \text{ с}^{-1}$

Продолжительность периода находится, примерно, в диапазоне 2300...2600 сек. Суммарное количество отказов за указанный период наблюдения не превышает 140.

При значительном росте интенсивности потока заявок при неизменной скорости обслуживания и числе мест в очереди мы наблюдаем рост потока отказов.

Если менять параметры интенсивности потока заявок на входе в СМО и число мест в очереди, то с использованием разработанной модели можно получить зависимости числа отказов от указанных параметров.

Результаты вычислительного эксперимента представлены в таблице 2.

Таблица №2. План и результаты моделирования.

$\lambda =$	<b>0,07</b>	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	<b>0,09</b>	Число отказов	20	2	0	0	0	0	0
$\lambda =$	<b>0,35</b>	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	<b>0,09</b>	Число отказов	188	170	158	152	142	126	117
$\lambda =$	<b>1,4</b>	Число мест в очереди	5	10	15	20	25	30	35
$\mu =$	<b>0,09</b>	Число отказов	189	181	170	166	162	148	142
	<b>Число отказов при относительном уровне отказов в СМО, %</b>	25,00%	50	50	50	50	50	50	50

Длина цепи событий в потоке – 200.

Графики зависимости числа отказов от указанных параметров мы видим на рис. 5.




Рисунок 5. Зависимость числа отказов от числа мест в очереди при разных интенсивностях входного потока заявок в СМО

Например, при данном объёме наблюдений относительная доля отказов на уровне 25% соответствует количеству отказов 50 (красная горизонтальная линия на диаграмме). Увеличение интенсивности потока заявок на входе в СМО в 20 раз с  $\lambda = 0,03 \text{ с}^{-1}$  до  $\lambda = 1,4 \text{ с}^{-1}$  приводит к резкому увеличению потока отказов, превышающему возможности системы по его обслуживанию. Число отказов при увеличении числа каналов от 1 до 7 находится в диапазоне 110...190, снижаясь незначительно, причём число отказов более, чем в 3 раза превышает 25% порог.

Модель воспроизводит эффект DOS-атаки на данную СМО.



### **Контрольные вопросы**

1. Дайте определение случайных процессов.
2. Объясните задачи и цели имитационного моделирования систем.
3. Метод Монте-Карло.
4. Как моделируют случайный процесс: опишите методы и алгоритмы моделирования
5. Что такое распределение случайных чисел.
6. Какие законы статистических распределений случайных величин Вы знаете?
7. Назовите основные характеристики статистических распределений: теоретическое определение и оценки по выборочным значениям.
8. Закон распределения Пуассона.
9. Экспоненциальный закон распределения.
10. Частоты распределения и гистограмма.
11. Опишите схему СМО с одним каналом и неограниченной ёмкостью накопителя.
12. Опишите схему СМО с неограниченным числом каналов без накопителя.