



МИНОБРНАУКИ РОССИИ

Федеральное государственное бюджетное образовательное учреждение
высшего образования

«МИРЭА – Российский технологический университет»

РТУ МИРЭА

Методические указания к выполнению практических работ

Моделирование информационно-аналитических систем

Практическая работа 9

	<i>(наименование дисциплины (модуля) в соответствии с учебным планом)</i>		
Уровень	специалитет		
	<i>(бакалавриат, магистратура, специалитет)</i>		
Форма обучения	очная		
	<i>(очная, очно-заочная, заочная)</i>		
Направление(-я) подготовки	10.05.04	Информационно-аналитические системы безопасности, специализации: специализация №1 "Автоматизация информационно-аналитической деятельности"; специализация №3 "Технологии информационно-аналитического мониторинга".	
	<i>(код(-ы) и наименование(-я))</i>		
Институт	Кибербезопасности и цифровых технологий		
	<i>(полное и краткое наименование)</i>		
Кафедра	Информационно-аналитические системы кибербезопасности (КБ-2)		
	<i>(полное и краткое наименование кафедры, реализующей дисциплину (модуль))</i>		
Лектор	к.т.н., доцент Лебедев Владимир Владимирович		
	<i>(сокращенно – ученая степень, ученое звание; полностью – ФИО)</i>		
Используются в данной редакции с учебного года	2024/25		
	<i>(учебный год цифрами)</i>		
Проверено и согласовано «___» _____ 20__ г.			
	<i>(подпись директора Института/Филиала с расшифровкой)</i>		

Москва 20__ г.

Практическое занятие №9

Численное имитационное моделирование стохастической системы обработки заявок на предоставление ресурса в автоматизированной информационной системе с отказами. Модели DOS-атак.

1. СМО с неограниченным числом каналов обслуживания без накопителя

Эта модель играет роль вспомогательной подзадачи для разработки модели системы с ограниченным числом каналов.

Схема СМО с параллельными каналами без накопителя представлена на рис.1.




Рисунок 1. Схема СМО без накопителя с параллельными каналами обслуживания

Граф модели Марковского процесса в этой системе представлен на рис. 2.

В отличие от модели СМО с одним каналом в этой схеме при подключении очередного канала скорость обслуживаниякратно возрастает. Каналы предоставляются транзактам независимо по мере их поступления в систему.




Рисунок 2. Граф Марковской модели СМО с неограниченным числом каналов.

Если число каналов обслуживания неограниченно, то все приходящие заявки обслуживаются в порядке поступления без возникновения отказов в обслуживании. Очередей в этой системе нет.

Задание.

1.1. Разработать имитационную событийно-временную модель процесса обслуживания транзактов в рассматриваемой СМО. Создать журнал наблюдения за событиями в потоке обслуживания.

1.2. Разработать имитационную математическую модель динамики изменения количества задействованных каналов во времени при заданных параметрах интенсивности поступления заявок в систему λ и интенсивности их обслуживания μ .

1.3. Построить диаграммы изменения количества каналов от времени процесса по временной шкале событий, связанных с занятием и освобождением каналов обслуживания.

Сделать выводы.

Пояснения к выполнению задания:

Схема дискретно-событийного процесса изменения количества заявок в рассматриваемой СМО представлена на рис. 3.




Рисунок 3. Модель занятия каналов в СМО без накопителя с неограниченным числом каналов без отказов

1. Сформировать временные шкалы событий в системе: 1-ая шкала – это случайные моменты времени занятия каналов; 2-ая шкала – это случайные моменты времени освобождения каналов.

Для моделирования временных шкал создать модель журнала событий (объём выборки 500 событий). Модель СМО создаётся по аналогии с моделью СМО с накопителем неограниченной ёмкости и одним каналом обслуживания, но при условии, что время ожидания для каждого транзакта равно 0 ($\Delta t_k^{ож} = 0$), потому что каждый транзакт входит в систему независимо от других и ему мгновенно предоставляется канал обслуживания.

В журнале фиксируются момент входа транзакта в систему и момент его выхода из системы по окончании обслуживания.

Пример решения задачи часть 1

Например, при параметрах процесса:

$\lambda =$	0,07
$\mu =$	0,03

Длина цепи событий в примере $N = 200$.

Сгенерированная модель журнала событий в стохастической системе СМО может иметь данные:

№	r_l	r_μ	$\Delta t_{\text{пост.}}$	$\Delta t_{\text{обсл.}}$	$t_{\text{пост.}}$	$\Delta t_{\text{ож.}}$	$t_{\text{вых.}}$	$\Delta t_{\text{пробы.}}$
1	0,3975	0,9922	7,2379	161,7740	7,2379	0	169,0119	161,7740
2	0,5313	0,5586	10,8257	27,2577	18,0636	0	45,3212	27,2577
3	0,2182	0,7831	3,5160	50,9409	21,5796	0	72,5205	50,9409
-----	-----	-----	-----	-----	-----	-----	-----	-----
198	0,3661	0,0863	6,5126	3,0100	2613,9406	0	2616,9505	3,0100
199	0,8106	0,4035	23,7707	17,2235	2637,7113	0	2654,9348	17,2235
200	0,6586	0,1011	15,3522	3,5533	2653,0634	0	2656,6167	3,5533

1-ая шкала – это данные столбца $t_{\text{пост}}$ таблицы, а 2-ая – это данные столбца $t_{\text{вых.}}$.

2. Расширить шкалы времени событий добавлением столбца с количеством “1” входящих в систему транзактов (равно числу занимаемых каналов) для 1-ой шкалы и ”-1” выходящих из системы транзактов (равно числу освобождаемых каналов).

Получим шкалы:

1-ая шкала	
$t_{\text{пост.}}$	Занимаемый канал
7,2379	1
18,0636	1
21,5796	1
35,2650	1
-----	-----
2607,4279	1
2613,9406	1
2637,7113	1
2653,0634	1

2-ая шкала	
$t_{\text{вых.}}$	Освобождаемый канал
169,0119	-1
45,3212	-1
72,5205	-1
68,0365	-1
-----	-----
2692,3017	-1
2616,9505	-1
2654,9348	-1
2656,6167	-1

Вид диаграммы потока событий в такой системе представлен на рис. 4.




Рисунок 4. Диаграмма потока событий в СМО с неограниченным числом каналов без накопителя

На диаграмме представлено 50 событий поступления заявок на предоставление канала и занятия канала транзактом в течение времени обслуживания.

Диаграмма наглядно демонстрирует независимость событий занятия каналов, времени обслуживания и пребывания транзактов в СМО рассматриваемого типа.

3. Объединить шкалы, например слиянием по принципу “одна над другой”:

№	Моменты времени	Выделяемые/освобождаемые каналы
1	7,2379	1
2	18,0636	1
3	21,5796	1
4	35,2650	1
	-----	-----
197	2607,4279	1
198	2613,9406	1
199	2637,7113	1
200	2653,0634	1
201	169,0119	-1
202	45,3212	-1
203	72,5205	-1
204	68,0365	-1
	-----	-----
397	2692,3017	-1
398	2616,9505	-1
399	2654,9348	-1
400	2656,6167	-1

4. Произвести сортировку таблицы по возрастанию значений столбца “моменты времени”.

5. Добавить справа к таблице столбец, в котором выполнить накопительное суммирование значений столбца “Выделяемые/освобождаемые каналы”, начиная с верхней ячейки.

В этом столбце будет отражаться текущее количество занятых каналов в системе.

На рисунке 5 представлена блок-схема алгоритма рассмотренной процедуры моделирования случайного дискретно-событийного процесса изменения количества обрабатываемых транзактов в СМО, и, следовательно, учёта изменения количества занимаемых каналов.




Рисунок 5. Блок-схема алгоритма модели процесса изменения количества обрабатываемых транзактов и занимаемых каналов в стохастической системе

Диаграмма процесса изменения количества транзактов, обрабатываемых в СМО, и занимаемых каналов в разные моменты времени объединённой шкалы событий будет иметь вид, см. рис. 6.

На диаграмме изображена красная линия ограничения числа каналов. В рассматриваемом примере число каналов равно 5.

Пики случайных значений числа заявок – требований на занятие каналов в некоторые моменты времени объединённой шкалы событий выходят над этой линией.

В эти моменты времени в системе с ограниченным числом каналов возникают события отказов, когда заявки на предоставление каналов приходящим в СМО транзактам отклоняются.




Рисунок 6. Диаграмма количества занимаемых каналов в разные моменты времени объединённой отсортированной шкалы событий

Если число каналов неограниченно, то поток отказов не возникает.

Объясните принцип учёта количества событий в потоке.

2. СМО с ограниченным числом каналов обслуживания без накопителя

Рассматривается модель системы массового обслуживания с непрерывным случайным экспоненциальным потоком заявок и непрерывным случайным экспоненциальным процессом их обслуживания в системе с параллельными каналами обслуживания, число которых не менее двух.

Граф модели Марковского процесса в такой СМО представлен ниже на рис. 7.




Рисунок 7. Граф Марковской модели СМО с ограниченным числом каналов и без ожидания

Если число каналов ограничено, то с определённой вероятностью в потоке событий может возникнуть событие, когда появление на входе в СМО заявки приходится на момент времени, когда все каналы заняты. В этом случае фиксируется **отказ** в предоставлении услуги, поскольку очередь ожидания отсутствует. Схема такого процесса представлена на рис. 8.




Рисунок 8. Модель процесса занятия каналов СМО без накопителя с отказами

Задание.

2.1. Разработать модель СМО с ограниченным числом каналов и отсутствием накопителя для образования мест в очереди ожидания обслуживания.

2.2. Построить зависимость количества отказов от времени при заданных параметрах модели: число каналов, интенсивность поступления заявок в систему λ и интенсивность их обслуживания μ .

2.3. Привести текст программы расчёта.

2.4. Построить диаграммы изменения количества отказов в обслуживании от времени процесса по временной шкале событий, связанных с занятием и освобождением каналов обслуживания.

2.5. Провести вычислительный эксперимент по исследованию зависимости числа отказов от интенсивности потока транзактов и числа каналов в СМО.

2.6. Провести планированный вычислительный эксперимент для построения регрессии, описывающей зависимость числа отказов от числа мест в очереди.

Сделать выводы.

Пояснения к выполнению задания.

1. Выполнить задание 1-й части практической работы, создав модель с неограниченным числом каналов.

Сформировать модель журнала событий длиной $N = 500$.

Сформировать таблицы шкал и построить табличную модель изменения количества обслуживаемых транзактов и занимаемых каналов в СМО без ограничений.

2. Задав конечное количество каналов (вариант задания) смоделировать поток отказов.

Модель построена с применением алгоритма расщепления событий в

сформированной шкале событий в первой модели без ограничений. При превышении суммы текущих заявок над количеством занятых каналов происходит фиксация событий потока отказов, при котором заявка, которая пришла систему, когда все каналы заняты, снимается с обслуживания.

Текста программы для моделирования рассмотренной процедуры и формирования табличной модели расчёта на дашборде витрины данных:

```
Private Sub Cr_un_shkala_sort ()
Dim K As Integer, Nch As Integer, Lyam, Mu, Nkr, Nk, ZZ
Dim A1(), A2(), A3(), A4(), A5(), A6(), A7(), A8()

'K - число заявок в цепи событий
'Nch - число каналов в СМО
'Lyam - интенсивность потока заявок
'Mu - интенсивность потока обслуживания

'Ввод исходных данных
K = InputBox("Ввести длину цепи событий в СМО")
Lyam = InputBox("Ввести интенсивность потока событий")
Mu = InputBox("Ввести интенсивность потока обслуживания")
Nch = InputBox("Ввести число каналов в СМО")
ReDim A1(1 To K, 1 To 2), A2(1 To K, 1 To 2), A3(1 To 2 * K, 1 To 3), A4(1 To 2 * K, 1 To 3)
ReDim A5(1 To 2 * K, 1 To 3), A6(1 To 2 * K, 1 To 3), A7(1 To 2 * K, 1 To 3), A8(1 To 2 * K, 1 To 3)

'Ввод данных временных параметров событий в СМО
'Формирование шкалы моментов времени поступления запросов в АИС
'Определение интервалов времени между поступлениями запросов в цепи событий
'Спецификация: Tin - время поступления в АИС; Tque - время поступления из очереди на обслуживание;
'Tout - время окончания обслуживания (выход из АИС); DTin - интервалы времени между поступлениями запросов в АИС в цепи событий;
'DTque - время задержки в очереди; DTserv - время обслуживания; DTout - время пребывания в системе
'A1 - шкала моментов поступления заявок на каналы в СМО
'A2 - шкала моментов времени освобождения каналов

'Ввод шкалы моментов времени поступления заявок на вход СМО
For J = 1 To K
Randomize
r = Rnd()
DTin = (1 / Lyam) * Log(1 / (1 - r))
If J = 1 Then
A1(J, 1) = DTin
Else
A1(J, 1) = A1(J - 1, 1) + DTin
End If
A1(J, 2) = 1
Next J

'Ввод шкалы моментов времени окончания обслуживания заявок в СМО
For J = 1 To K
Randomize
```

```

        r = Rnd()
        DTserv = (1 / Mu) * Log(1 / (1 - r))
        A2(J, 1) = A1(J, 1) + DTserv
        A2(J, 2) = -1
    Next J

'Формирование сводной шкалы событий A1-A2
    For M = 1 To K
        A3(2 * M - 1, 1) = A1(M, 1)
        A3(2 * M - 1, 2) = A1(M, 2)
        A3(2 * M - 1, 3) = M
        A3(2 * M, 1) = A2(M, 1)
        A3(2 * M, 2) = A2(M, 2)
        A3(2 * M, 3) = M
    Next M

'Сортировка сводной шкалы событий
    For J = 1 To 2 * K
        A4(J, 1) = A3(J, 1)
        A4(J, 2) = A3(J, 2)
        A4(J, 3) = A3(J, 3)
    Next J
    KK = 1
    Do While KK <= 2 * K
        For J = KK + 1 To 2 * K
            If A4(J, 1) < A4(KK, 1) Then
                AA = A4(KK, 1)
                BB = A4(KK, 2)
                CC = A4(KK, 3)
                A4(KK, 1) = A4(J, 1)
                A4(KK, 2) = A4(J, 2)
                A4(KK, 3) = A4(J, 3)
                A4(J, 1) = AA
                A4(J, 2) = BB
                A4(J, 3) = CC
            End If
        Next J
        KK = KK + 1
    Loop

'Моделирование динамики занятия и освобождения каналов и формирование потока
отказов
'Динамика требований на занятие каналов --> A5
    Nk = 0
    For M = 1 To 2 * K
        Nk = Nk + A4(M, 2)
        A5(M, 1) = A4(M, 1)
        A5(M, 2) = Nk
        A5(M, 3) = A4(M, 3)
    Next M

'Модифицированная шкала событий при ограниченном числе каналов в СМО -->A7 и
шкала потока отказов --> A6
    For M = 1 To 2 * K
        A7(M, 1) = A4(M, 1)
        A7(M, 2) = A4(M, 2)
        A7(M, 3) = A4(M, 3)
    Next M
    Z = 0
    A6(1, 1) = A7(1, 1)
    A6(1, 2) = 0
    A6(1, 3) = 0
    For M = 2 To 2 * K

```

```

Nkr = 0
For J = 1 To M
    Nkr = Nkr + A7(J, 2)
Next J
If Nkr > Nch Then
    A6(M, 2) = 1
    A6(M, 1) = A7(M, 1)
    A6(M, 3) = A7(M, 3)
    A7(M, 2) = 0
    Z = A7(M, 3)
    n = M
    E = 0
    Do While E = 0
        n = n + 1
        If A7(n, 3) = Z Then
            T = n
            E = 1
        End If
    Loop
    A7(T, 2) = 0
Else
    A6(M, 2) = 0
    A6(M, 1) = A7(M, 1)
    A6(M, 3) = 0
    If A4(M, 3) = Z Then
        A7(M, 2) = 0
    End If
    T = M
End If
Next M

'Динамика занятия каналов в СМО с ограниченным числом каналов --> A8
Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A7(M, 2)
    A8(M, 1) = A7(M, 1)
    A8(M, 2) = Nk
    A8(M, 3) = A7(M, 3)
Next M

'Вывод результатов расчёта
ncol = 2
nstr = 20

Ndenide = 0
For J = 1 To 2 * K
    Ndenide = Ndenide + A6(J, 2)
    Cells(nstr + K + 5 + J, ncol + 15) = Ndenide
Next J

Cells(3, 1) = "Длина цепи событий"
Cells(3, 2) = K
Cells(5, 1) = "Lyam"
Cells(6, 1) = "Mu"
Cells(5, 2) = Lyam
Cells(6, 2) = Mu
Cells(nstr, ncol) = "№"
Cells(nstr, ncol + 1) = "Tin"
Cells(nstr, ncol + 2) = "Tout"
Cells(nstr, ncol + 3) = "DTin"
Cells(nstr, ncol + 4) = "DTwait"
Cells(nstr, ncol + 5) = "DTlife"

```

```

For M = 1 To K
    Cells(nstr + M, ncol) = M
    Cells(nstr + M, ncol + 1) = A1(M, 1)
    Cells(nstr + M, ncol + 2) = A2(M, 1)
    If M = 1 Then
        Cells(nstr + M, ncol + 3) = A1(M, 1)
    Else
        Cells(nstr + M, ncol + 3) = A1(M, 1) - A1(M - 1, 1)
    End If
    Cells(nstr + M, ncol + 4) = 0
    Cells(nstr + M, ncol + 5) = A2(M, 1) - A1(M, 1)
Next M

Cells(nstr + K + 5, ncol) = "№"
Cells(nstr + K + 5, ncol + 1) = "Tin"
Cells(nstr + K + 5, ncol + 2) = "n+"
Cells(nstr + K + 5, ncol + 3) = "Tout"
Cells(nstr + K + 5, ncol + 4) = "n-"
Cells(nstr + K + 5, ncol + 5) = "№"
Cells(nstr + K + 5, ncol + 6) = "UnSc"
Cells(nstr + K + 5, ncol + 7) = "n+/-"
Cells(nstr + K + 5, ncol + 8) = "SortUnSc"
Cells(nstr + K + 5, ncol + 9) = "n+/-"
Cells(nstr + K + 5, ncol + 10) = "n_Din"
Cells(nstr + K + 5, ncol + 11) = "Nlim"
Cells(nstr + K + 5, ncol + 12) = "n_Din_lim"
Cells(nstr + K + 5, ncol + 13) = "n_Din_denide"
Cells(nstr + K + 5, ncol + 14) = "№_Din_denide"
Cells(nstr + K + 5, ncol + 15) = "Sum_D_denide"
Cells(nstr + K + 5, ncol + 16) = "Number_denide"
Cells(nstr + K + 5 + 1, ncol + 16) = Ndenide

For M = 1 To K
    Cells(nstr + K + 5 + M, ncol) = M
    Cells(nstr + K + 5 + M, ncol + 1) = A1(M, 1)
    Cells(nstr + K + 5 + M, ncol + 2) = A1(M, 2)
    Cells(nstr + K + 5 + M, ncol + 3) = A2(M, 1)
    Cells(nstr + K + 5 + M, ncol + 4) = A2(M, 2)
Next M

For M = 1 To 2 * K
    Cells(nstr + K + 5 + M, ncol + 5) = M
    Cells(nstr + K + 5 + M, ncol + 6) = A3(M, 1)
    Cells(nstr + K + 5 + M, ncol + 7) = A3(M, 2)
    Cells(nstr + K + 5 + M, ncol + 8) = A4(M, 1)
    Cells(nstr + K + 5 + M, ncol + 9) = A4(M, 2)
    Cells(nstr + K + 5 + M, ncol + 10) = A5(M, 2)
    Cells(nstr + K + 5 + M, ncol + 11) = Nch
    Cells(nstr + K + 5 + M, ncol + 12) = A8(M, 2)
    Cells(nstr + K + 5 + M, ncol + 13) = A6(M, 2)
    Cells(nstr + K + 5 + M, ncol + 14) = A6(M, 3)
Next M
End Sub

```

Продолжение примера решения задачи часть 2

Если число каналов в СМО с рассмотренными в предыдущем задании параметрами ограничено, допустим числом 5, то на диаграмме рис. 6 это представлено красной горизонтальной линией. В моменты времени, когда число заявок – требований на занятие каналов превышает число 5, возникает поток отказов, см. рис. 7. В данном случае при интенсивности потока заявок на входе в СМО $\lambda = 0,07 \text{ с}^{-1}$ и интенсивности потока обработки заявок в каналах $\mu = 0,03 \text{ с}^{-1}$ поток отказов, красные столбики на диаграмме рис. 7, незначительный. Диаграмма накопления числа отказов за период наблюдения за процессом в пределах 200 заявок на входе в СМО представлена на рис. 8.

Продолжительность периода находится, примерно, в диапазоне 2700...2800 сек. Суммарное количество отказов за указанный период наблюдения не превышает 10.

При значительном росте интенсивности потока заявок при неизменной скорости обслуживания и числе каналов мы будем наблюдать рост потока отказов.



Рисунок 7. Диаграмма количества занимаемых каналов (не больше 5) – синие столбики и диаграмма количества отказов – красные столбики за наблюдаемый период вхождения в СМО 200 заявок при интенсивности потока заявок $\lambda = 0,07 \text{ с}^{-1}$ и интенсивности потока обслуживания $\mu = 0,03 \text{ с}^{-1}$




Рисунок 8. Диаграмма накопления числа отказов за период вхождения в СМО 200 заявок при интенсивности потока заявок $\lambda = 0,07 \text{ с}^{-1}$ и интенсивности потока обслуживания $\mu = 0,03 \text{ с}^{-1}$

Результаты вычислительного эксперимента (п. 2.5.)

Если менять параметры интенсивности потока заявок на входе в СМО и число каналов, то с использованием разработанной модели можно получить зависимости числа отказов от указанных параметров.

План вычислительного эксперимента представлен в таблице 1.

Для вычисления количества отказов, представляющих собой отклики на изменение объясняющих факторов, мы используем подпрограмму-функцию, полученную путём модификации программы имитационной модели процесса, рассмотренной ранее.

Таблица 1. План эксперимента

$\lambda=$	λ_B	Число каналов	1	2	3	4	5	6	7	8	9	10
$\mu=$	μ_B	Число отказов										
$\lambda=$	$5\lambda_B$	Число каналов	1	2	3	4	5	6	7	8	9	10
$\mu=$	μ_B	Число отказов										
$\lambda=$	$20\lambda_B$	Число каналов	1	2	3	4	5	6	7	8	9	10
$\mu=$	μ_B	Число отказов										
Число отказов при относительном уровне отказов в СМО, %		5,00%	25	25	25	25	25	25	25	25	25	25

Результаты эксперимента по условиям рассматриваемого примера представлены в таблице 2.

Таблица №2. План и результаты моделирования

$\lambda=$	0,07	Число каналов	1	2	3	4	5	6	7
$\mu=$	0,09	Число отказов	72	19	13	3	0	0	0
$\lambda=$	0,35	Число каналов	1	2	3	4	5	6	7
$\mu=$	0,09	Число отказов	151	132	87	67	28	23	11
$\lambda=$	1,4	Число каналов	1	2	3	4	5	6	7
$\mu=$	0,09	Число отказов	188	171	162	139	133	131	112
Число отказов при относительном уровне отказов в СМО, %		25,00%	50	50	50	50	50	50	50

Графики зависимости числа отказов от указанных параметров мы видим на рис. 9.



Рисунок 9. Зависимость числа отказов от числа каналов при разных интенсивностях входного потока заявок в СМО

Например, при данном объёме наблюдений относительная доля отказов на уровне 25% соответствует количеству отказов 50 (красная горизонтальная линия на диаграмме). Увеличение интенсивности потока заявок на входе в СМО в 20 раз с $\lambda = 0,03 \text{ с}^{-1}$ до $\lambda = 1,4 \text{ с}^{-1}$ приводит к резкому увеличению потока отказов, превышающему возможности системы по его обслуживанию. Число отказов при увеличении числа каналов от 1 до 7 находится в диапазоне 110...190, снижаясь незначительно, причём число отказов более, чем в 3 раза превышает 25% порог.

Модель воспроизводит эффект DOS-атаки на данную СМО.

Сделать выводы по результатам моделирования.

Текст подпрограммы-функции:

```
Function Dens_Numbs_(X As Integer, Y, Z, W As Integer)
Dim K As Integer, Nch As Integer, Lyam, Mu, Nkr, Nk, ZZ
Dim A1(), A2(), A3(), A4(), A5(), A6(), A7(), A8()
'K - число заявок в цепи событий
'Nch - число каналов в СМО
'Lyam - интенсивность потока заявок
'Mu - интенсивность потока обслуживания
K = X
Lyam = Y
Mu = Z
Nch = W
ReDim A1(1 To K, 1 To 2), A2(1 To K, 1 To 2), A3(1 To 2 * K, 1 To 3), A4(1 To 2 * K, 1 To 3)
```

```

ReDim A5(1 To 2 * K, 1 To 3), A6(1 To 2 * K, 1 To 3), A7(1 To 2 * K, 1 To 3),
A8(1 To 2 * K, 1 To 3)
'Ввод данных временных параметров событий в СМО
'Формирование шкалы моментов времени поступления запросов в АИС
'Определение интервалов времени между поступлениями запросов в цепи событий
'Спецификация: Tin - время поступления в АИС; Tque - время поступления из
очереди на обслуживание;
'Tout - время окончания обслуживания (выход из АИС); DTin - интервалы времени
между поступлениями запросов в АИС в цепи событий;
'DTque - время задержки в очереди; DTserv - время обслуживания; DTout - время
пребывания в системе

'Ввод шкалы моментов времени поступления заявок на вход СМО
'A1 - шкала моментов поступления заявок на каналы в СМО

    For J = 1 To K
        Randomize
        r = Rnd()
        DTin = (1 / Lyam) * Log(1 / (1 - r))
        If J = 1 Then
            A1(J, 1) = DTin
        Else
            A1(J, 1) = A1(J - 1, 1) + DTin
        End If
        A1(J, 2) = 1
    Next J

'Ввод шкалы моментов времени окончания обслуживания заявок в СМО
'A2 - шкала моментов времени освобождения каналов

    For J = 1 To K
        Randomize
        r = Rnd()
        DTserv = (1 / Mu) * Log(1 / (1 - r))
        A2(J, 1) = A1(J, 1) + DTserv
        A2(J, 2) = -1
    Next J

'Формирование сводной шкалы событий A1-A2

    For M = 1 To K
        A3(2 * M - 1, 1) = A1(M, 1)
        A3(2 * M - 1, 2) = A1(M, 2)
        A3(2 * M - 1, 3) = M
        A3(2 * M, 1) = A2(M, 1)
        A3(2 * M, 2) = A2(M, 2)
        A3(2 * M, 3) = M
    Next M

'Сортировка сводной шкалы событий

    For J = 1 To 2 * K
        A4(J, 1) = A3(J, 1)
        A4(J, 2) = A3(J, 2)
        A4(J, 3) = A3(J, 3)
    Next J

    KK = 1
    Do While KK <= 2 * K
        For J = KK + 1 To 2 * K
            If A4(J, 1) < A4(KK, 1) Then
                AA = A4(KK, 1)
                BB = A4(KK, 2)

```

```

        CC = A4(KK, 3)
        A4(KK, 1) = A4(J, 1)
        A4(KK, 2) = A4(J, 2)
        A4(KK, 3) = A4(J, 3)
        A4(J, 1) = AA
        A4(J, 2) = BB
        A4(J, 3) = CC
    End If
Next J
KK = KK + 1
Loop

'Моделирование динамики занятия и освобождения каналов и формирование потока
отказов
'Динамика требований на занятие каналов

Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A4(M, 2)
    A5(M, 1) = A4(M, 1)
    A5(M, 2) = Nk
    A5(M, 3) = A4(M, 3)
Next M

'Модифицированная шкала событий при ограниченном числе каналов в СМО -->A7 и
шкала потока отказов --> A6

For M = 1 To 2 * K
    A7(M, 1) = A4(M, 1)
    A7(M, 2) = A4(M, 2)
    A7(M, 3) = A4(M, 3)
Next M
ZZ = 0
A6(1, 1) = A7(1, 1)
A6(1, 2) = 0
A6(1, 3) = 0
For M = 2 To 2 * K
    Nkr = 0
    For J = 1 To M
        Nkr = Nkr + A7(J, 2)
    Next J
    If Nkr > Nch Then
        A6(M, 2) = 1
        A6(M, 1) = A7(M, 1)
        A6(M, 3) = A7(M, 3)
        A7(M, 2) = 0
        ZZ = A7(M, 3)
        n = M
        E = 0
        Do While E = 0
            n = n + 1
            If A7(n, 3) = ZZ Then
                T = n
                E = 1
            End If
        Loop
        A7(T, 2) = 0
    Else
        A6(M, 2) = 0
        A6(M, 1) = A7(M, 1)
        A6(M, 3) = 0
        If A4(M, 3) = ZZ Then
            A7(M, 2) = 0
        End If
    End If
Next M

```

```

End If
T = M
End If
Next M

'Динамика занятия каналов в СМО с ограниченным числом каналов
'A8 - поток занятия каналов в моменты времени событий объединённой сортированной
шкалы

Nk = 0
For M = 1 To 2 * K
    Nk = Nk + A7(M, 2)
    A8(M, 1) = A7(M, 1)
    A8(M, 2) = Nk
    A8(M, 3) = A7(M, 3)
Next M

'Суммарное количество отказов за время наблюдения
DenSum = 0
For J = 1 To 2 * K
    DenSum = DenSum + A6(J, 2)
Next J

Dens_Numbs_ = DenSum

End Function

```

Проведение вычислительного эксперимента (п. 2.6.)

Параметры модели задать по варианту:

$\lambda =$	$5\lambda_B$
$\mu =$	μ_B

План эксперимента принять согласно таблице 3. Использовать 5 серий параллельных экспериментов.

Таблица 3. План эксперимента для построения регрессии

$\lambda =$	$5\lambda_B$	Число каналов										
$\mu =$	μ_B		1	2	3	4	5	6	7	8	9	10
Серии		1										
		2										
		3										
		4										
		5										
		Среднее по сериям										

Результаты эксперимента по условиям рассматриваемого примера представлены в таблице 4.

Таблица 4. План и результаты эксперимента для построения регрессии

$\lambda=$	0,09	Число каналов	1	2	3	4	5	6	7	8	9	10
$\mu=$	0,03	1	385	241	189	102	44	23	14	6	0	2
		2	382	270	182	103	65	31	0	0	0	0
		3	381	259	145	112	58	24	10	4	0	0
		4	369	275	172	100	61	31	15	4	0	0
		5	390	249	177	86	65	37	14	2	4	0
		Среднее по сериям	381,4	258,8	173	100,6	58,6	29,2	10,6	3,2	0,8	0,4

График зависимости числа отказов от числа каналов по последней строке таблицы 4 мы видим на рис. 10.




Рисунок 10. График зависимости среднего числа отказов от числа каналов СМО

Усреднение случайных значений откликов эксперимента позволяет получить более сглаженные данные, усреднённые по большему объёму данных.

Вычисление откликов производим с использованием подпрограммы функции, представленной выше.

В качестве модели использовать параметрическую функцию вида:

$$y = N \cdot e^{-ax^b}.$$

В формуле: y – среднее число отказов; x – число каналов в СМО; N – предельное число отказов в СМО, равное числу событий, рассматриваемых в модели; a, b – параметры модели регрессии.

В данной системе при отсутствии каналов обслуживание транзактов невозможно, поэтому предельное состояние есть 100% отказ.

Моделирование регрессионных зависимостей из практики 8 и 9 произведём в практике 10, тема которой посвящена статистической обработке данных.

Контрольные вопросы

1. Дайте определение случайных процессов.
2. Объясните задачи и цели имитационного моделирования систем.
3. Метод Монте-Карло.
4. Как моделируют случайный процесс: опишите методы и алгоритмы моделирования
5. Что такое распределение случайных чисел.
6. Какие законы статистических распределений случайных величин Вы знаете?
7. Назовите основные характеристики статистических распределений: теоретическое определение и оценки по выборочным значениям.
8. Закон распределения Пуассона.
9. Экспоненциальный закон распределения.
10. Частоты распределения и гистограмма.
11. Опишите схему СМО с одним каналом и неограниченной ёмкостью накопителя.
12. Опишите схему СМО с неограниченным числом каналов без накопителя.

СПИСОК ИСТОЧНИКОВ И ЛИТЕРАТУРЫ

1. Дорошенко А.Н. Математическое и имитационное моделирование дискретных процессов и систем [Электронный ресурс]: учебное пособие /А. Н. Дорошенко. — М.: МИРЭА, 2018. — 151 с. Электрон, опт. диск (ISO)
2. В.В. Лозовецкий. Защита автоматизированных систем обработки информации и телекоммуникационных сетей: учебное пособие для вузов/ В.В. Лозовецкий, Е.Г. Комаров, В.В. Лебедев; под редакцией В.В. Лозовецкого. – Санкт-Петербург: Лань, 2023, -448 с: ил. – Текст: непосредственный.
3. Ермакова А.Ю. Моделирование автоматизированных систем в защищённом исполнении [Электронный ресурс]: Учебное пособие, ч.1./ Ермакова А.Ю., Лебедев В.В. — М.: МИРЭА – Российский технологический университет, 2024. — 1 электрон. опт. диск (CD-ROM)..
4. Пестриков В.М., Дудкин В.С., Петров Г.А.. Дискретная математика./Уч. пос.. – СПб.: СПб ГТУРП, 2013.- 136 с.
5. Гельгор А.Л., Горлов А.И., Попов Е.А.. Методы моделирования случайных величин и случайных процессов: уч. пос. – СПб.: Изд-во ПГПУ, 2012. — 217 с.
6. Васильев К.К., Служивый М.Н. Математическое моделирование систем связи: учеб, пособие. — УлГТУ, 2008 — 168 с.
7. Карпов Ю.Г. Имитационное моделирование систем. Введение в моделирование с AnyLogic 5. — СПб.: БХВ - Петербург, 2006. — 400 с.
8. Полянский Д.И. Оценка защищённости./Уч. пос. – Владимир: изд-во Владим. гос. ун-та, 2005. – 80 с.
9. Шмидт Б. Введение в имитационное моделирование в системе Simplex3 / пер. с нем. Ю.А. Ивашкина. — М.: Наука, 2003. — 30 с.
10. Шмидт Б. Искусство моделирования и имитации. Введение в имитационную систему Simplex3 : пер. с нем.: SCS-Европа BVBA, Гент. Бельгия. 2003. — 550 с.

11. Харин Ю.С. и др. Основы имитационного и статистического моделирования. Учебное пособие – Мн.: Дизайн ПРО, 1997. – 288 с.
12. Шеннон Р. Имитационное моделирование систем - искусство и наука: Пер. с англ. – М.: Мир, 1978.